
Subject: Re: Inverse FFT

Posted by [R.G. Stockwell](#) on Tue, 17 Dec 2002 15:02:57 GMT

[View Forum Message](#) <> [Reply to Message](#)

Here is an (main level) example that hopefully does what you want. Note the slightly better precision of the fft method, due to the superior method of calculating the same thing. Also, there is a huge difference in speed, especially as N gets larger than say 10 or so. On my computer 1.6Ghz athlon, the times are about a factor of 20. (i.e. fft 20 times faster than "by hand"). The point I am getting to is "don't inverse fft by hand".

And note that I left the resulting arrays as complex, but it is equal to the original time series, since the imaginary part is zero. You may want to cast them to float (or double)

Cheers,
bob stockwell

```
; make a time series
len = 16
a = randomn(seed,len)
a = double(a)

; calc spectrum
ft = fft(a)

; inverse by fft FAST!
tic = systime(1)
ift = fft(ft,/inverse)
toc = systime(1)
print,'fft time = ', (toc - tic)*1000d ; microseconds

; inverse by hand SLOW!
tic = systime(1)
byhand = dcomplexarr(len)
t = dindgen(len)
for i = 0,len-1 do begin
  byhand = byhand + ft[i]*exp(complex(0,1)*2*!dpi*t*i/len)
endfor
toc = systime(1)
print,'by hand time = ', (toc - tic)*1000d

; print out results
print,'original time series'
```

```
print,a
print,'inverse by fft'
print,ift
print,'inverse by hand'
print,byhand

; plot out results
!p.multi = [ 0,1,3]
plot,a,tit='timeseries'
plot,ift,tit='by fft'
oplot,imaginary(ift),linestyle=2
plot,byhand,tit='byhand'
oplot,imaginary(byhand),linestyle=2
end
```
